


6. 応答の数値積分

- 地震動のような不規則な外乱を受ける構造物の動的応答解析には、微小な時間間隔ごとに運動方程式を数値的に積分して解を求める数値積分法を用いることが多い。




2018/4/26

建築数理基礎論

43

数値積分法の種類

- 加速度法
 - 最もよく用いられている
 - 線形加速度法
 - 平均加速度法
 - 段階加速度法
 - 衝撃加速度法
- Runge-Kutta法
- 厳密解に基づく方法



2018/4/26

建築数理基礎論

44

線形加速度法

$$\ddot{y}(t) = \ddot{y}_n + \frac{\ddot{y}_{n+1} - \ddot{y}_n}{\Delta t} (t - t_n) \quad \text{1次式}$$


$$\dot{y}(t) = \dot{y}_n + \ddot{y}_n (t - t_n)$$

$$+ \frac{\ddot{y}_{n+1} - \ddot{y}_n}{2\Delta t} (t - t_n)^2 \quad \text{2次式}$$

$$y(t) = y_n + \dot{y}_n (t - t_n) + \frac{1}{2} \ddot{y}_n (t - t_n)^2$$

$$+ \frac{\ddot{y}_{n+1} - \ddot{y}_n}{6\Delta t} (t - t_n)^3 \quad \text{3次式}$$

$t - t_n = \Delta t$ を代入すると \dot{y}_{n+1} , y_{n+1} が求まる



2018/4/26

建築数理基礎論

45

線形加速度法の解

$m\ddot{y} + c\dot{y} + ky = -m\ddot{y}_0$ より

$$\ddot{y}_{n+1} = -\frac{c}{m} \dot{y}_{n+1} - \frac{k}{m} y_{n+1} - \ddot{y}_{0n+1} \quad \dots \textcircled{1}$$

$$\dot{y}_{n+1} = \dot{y}_n + \frac{\ddot{y}_n + \ddot{y}_{n+1}}{2} \Delta t \quad \dots \textcircled{2}$$

$$y_{n+1} = y_n + \dot{y}_n \Delta t + \frac{2\ddot{y}_n + \ddot{y}_{n+1}}{6} \Delta t^2 \quad \dots \textcircled{3}$$

未知数が3個なので、①、②、③を連立してとけば、解が求まる

$$\ddot{y}_{n+1} = -\frac{\ddot{y}_{0n+1} + \frac{c}{m} \left(\dot{y}_n + \frac{1}{2} \ddot{y}_n \Delta t \right) + \frac{k}{m} \left(y_n + \dot{y}_n \Delta t + \frac{1}{3} \ddot{y}_n \Delta t^2 \right)}{1 + \frac{1}{2} \frac{c}{m} \Delta t + \frac{1}{6} \frac{k}{m} \Delta t^2}$$

2018/4/26

建築数理基礎論

46

増分変位 Δy による解(線形加速度法)

$$\Delta \ddot{y} = -\frac{c}{m} \Delta \dot{y} - \frac{k}{m} \Delta y - \Delta \ddot{y}_0 \quad \dots \textcircled{1}$$

$$\Delta \dot{y} = \dot{y}_{n+1} - \dot{y}_n = \frac{\ddot{y}_n + \ddot{y}_{n+1}}{2} \Delta t = \ddot{y}_n \Delta t + \frac{1}{2} \Delta \ddot{y} \Delta t^2 \quad \dots \textcircled{2}$$

$$\Delta y = y_{n+1} - y_n = \dot{y}_n \Delta t + \frac{1}{2} \ddot{y}_n \Delta t^2 + \frac{1}{6} \Delta \ddot{y} \Delta t^3 \quad \dots \textcircled{3}$$

①、②、③を連立して解くと

$$\Delta y = \Delta \bar{P} / \bar{k}$$

$$\text{ただし、} \Delta \bar{P} = m \left(-\Delta \ddot{y}_0 + \frac{6}{\Delta t} \dot{y}_n + 3\ddot{y}_n \right) + c \left(3\dot{y}_n + \frac{\Delta t}{2} \ddot{y}_n \right)$$

$$\bar{k} = k + \frac{3}{\Delta t} c + \frac{6}{\Delta t^2} m$$

$$\Delta \dot{y} = \frac{3}{\Delta t} \Delta y - 3\dot{y}_n - \frac{\Delta t}{2} \ddot{y}_n$$

$$\Delta \ddot{y} = \frac{6}{\Delta t^2} \Delta y - \frac{6}{\Delta t} \dot{y}_n - 3\ddot{y}_n$$

2018/4/26

建築数理基礎論

47

平均加速度法


$$\ddot{y}_{n+1} = -\frac{c}{m} \dot{y}_{n+1} - \frac{k}{m} y_{n+1} - \ddot{y}_{0n+1}$$

$$\dot{y}_{n+1} = \dot{y}_n + \frac{\ddot{y}_n + \ddot{y}_{n+1}}{2} \Delta t$$

$$y_{n+1} = y_n + \dot{y}_n \Delta t + \frac{\ddot{y}_n + \ddot{y}_{n+1}}{4} \Delta t^2$$

これらを解くと

$$\ddot{y}_{n+1} = -\frac{\ddot{y}_{0n+1} + \frac{c}{m} \left(\dot{y}_n + \frac{1}{2} \ddot{y}_n \Delta t \right) + \frac{k}{m} \left(y_n + \dot{y}_n \Delta t + \frac{1}{4} \ddot{y}_n \Delta t^2 \right)}{1 + \frac{1}{2} \frac{c}{m} \Delta t + \frac{1}{4} \frac{k}{m} \Delta t^2}$$



2018/4/26

建築数理基礎論

48

増分変位 Δy による解(平均加速度法)

$$\Delta y = \overline{\Delta P} / \bar{k}$$

ただし、 $\overline{\Delta P} = m \left(-\Delta \ddot{y}_0 + \frac{4}{\Delta t} \dot{y}_n + 2\ddot{y}_n \right) + 2c\dot{y}_n$

$$\bar{k} = k + \frac{2}{\Delta t} c + \frac{4}{\Delta t^2} m$$

$$\Delta \dot{y} = \frac{2}{\Delta t} \Delta y - 2\dot{y}_n$$

$$\Delta \ddot{y} = \frac{4}{\Delta t^2} \Delta y - \frac{4}{\Delta t} \dot{y}_n$$

2018/4/26

建築数理基礎論

49

Newmarkの β 法

- 係数 β を用いると加速度法の数値積分の公式をまとめて記述できる

$$\Delta y = \overline{\Delta P} / \bar{k}$$

ただし、 $\overline{\Delta P} = m \left(-\Delta \ddot{y}_0 + \frac{1}{\beta \Delta t} \dot{y}_n + \frac{1}{2\beta} \ddot{y}_n \right) + c \left(\frac{1}{2\beta} \dot{y}_n + \left(\frac{1}{4\beta} - 1 \right) \ddot{y}_n \Delta t \right)$

$$\bar{k} = k + \frac{1}{2\beta \Delta t} c + \frac{1}{\beta \Delta t^2} m$$

$$\Delta \dot{y} = \frac{1}{2\beta \Delta t} \Delta y - \frac{1}{2\beta} \dot{y}_n - \left(\frac{1}{4\beta} - 1 \right) \ddot{y}_n \Delta t$$

$$\Delta \ddot{y} = \frac{1}{\beta \Delta t^2} \Delta y - \frac{1}{\beta \Delta t} \dot{y}_n - \frac{1}{2\beta} \ddot{y}_n$$

平均加速度法 $\beta=1/4$
線形加速度法 $\beta=1/6$
段階加速度法 $\beta=1/8$
衝撃加速度法 $\beta=0$

2018/4/26

建築数理基礎論

50

Newmarkの β 法の安定性

- 数値積分の時間刻み Δt が固有周期 T に対してある程度以上大きいと、発散することがある。

- $\beta \geq 1/4$ の時 Δt に関わらず安定
- $\beta < 1/4$ の時 $\Delta t \leq \frac{1}{\pi \sqrt{1-4\beta}} T$ なら安定

一般的には、時間刻み $\Delta t < T/8$ 程度として計算するのが良い

2018/4/26

建築数理基礎論

51

1質点系の線形応答解析プログラム

SDF_elastic

- 機能:

- 1質点系の振動解析(自由振動、地震応答)
- 解析結果のファイルへの出力
- 変位の時刻歴のグラフ表示
- 質点の変位のアニメーション表示

- 使用言語: Visual Basic ver.6

2018/4/26

建築数理基礎論

52

プログラムの構成

- フォーム: SDF_elastic.frm

プログラムの主要部

- 解析モデルの設定
- 地震加速度記録の読み込み
- 応答計算(数値積分)
- グラフ、アニメーションの表示
- 最大応答の抽出
- データのファイルへの出力

- 標準モジュール: 変数定義.bas


プログラムで使用する変数、定数の定義をまとめて記述

2018/4/26

建築数理基礎論

53

プログラムのフロー



2018/4/26

建築数理基礎論

54

Option Explicit

```
Private Sub Form_Load()
    Timer1.Interval = 10
End Sub
```

```
Private Sub Start_Click()
```

```
'*** 解析モデルの設定
    T = Val(txtT.Text)           ' 周期
    M = 1#                       ' 質量
    K = (2 * 3.14159265 / T) ^ 2 * M ' 剛性
    H = Val(txtH.Text) / 100     ' 減衰定数
    C = 2 * H * Sqr(K / M)
    bet = 1 / Val(txtBeta.Text)

'*** 地震記録の読み込み
    If Opt解析2.Value = True Then
        Call 地震記録
    Else
        dT = 0.01
        Tend = Val(txtTend.Text)
        Tstep = Int(Tend / dT + 0.5)
        For i = 0 To Tstep
            Acc(i) = 0
        Next i
    End If

'*** データの初期化
    X0 = Val(txtX0.Text) / 100: X = 0 ' 変位
    V0 = Val(txtV0.Text) / 100: V = 0 ' 速度
    A0 = -X0 * (K / M): A = 0        ' 加速度
    Xmax = 0: Xmin = 0
    Vmax = 0: Vmin = 0
    Amax = 0: Amin = 0

    Call グラフ初期設定
    Call 質点初期設定
    If opt保存2.Value = True Then
        Call 出力ファイル初期設定
    End If

    経過時間 = 0#

'*** 応答計算のメインループ
    For istep = 1 To Tstep
        経過時間 = 0#

        Call 数値積分
        Call グラフ
        Call 質点アニメ
        If opt保存2.Value = True Then
            Call データ出力
        End If

        X0 = X: V0 = V: A0 = A

'*** 最大・最小応答値の抽出
        If Xmax < X Then Xmax = X
        If Xmin > X Then Xmin = X
        If Vmax < V Then Vmax = V
        If Vmin > V Then Vmin = V
        If Amax < A + Acc(istep) Then Amax = A + Acc(istep)
        If Amin > A + Acc(istep) Then Amin = A + Acc(istep)

'*** 応答結果の表示
        txtTime.Text = istep * dT
        txtDisp.Text = Format(X * 100, "##0.000")
        txtVel.Text = Format(V * 100, "##0.00")
        txtAcc.Text = Format((A + Acc(istep)) * 100, "###0.0")
        txtXmax.Text = Format(Xmax * 100, "##0.000")
        txtXmin.Text = Format(Xmin * 100, "##0.000")
        txtVmax.Text = Format(Vmax * 100, "##0.00")
        txtVmin.Text = Format(Vmin * 100, "##0.00")
        txtAmax.Text = Format(Amax * 100, "##0.0")
        txtAmin.Text = Format(Amin * 100, "##0.0")

'*** 描画の時間刻みを実時間にあわせる
100     dmy = DoEvents()
        If 経過時間 < dT Then GoTo 100
    Next istep
End Sub
```

```
*****
```

```
Private Sub 地震記録()
```

```
*****
```

```
On Error GoTo OpenError
```

```
EqFile = Dir1.Path & "¥" & File1.FileName '地震波ファイル名
```

```
Open EqFile For Input As #1
```

```
Line Input #1, EqName
```

```
txtEqName.Text = EqName
```

```
Input #1, dT, Unit, A0max
```

```
' * 解析ステップ数の計算
```

```
Tend = Val(txtTend.Text)
```

```
Tstep = Int(Tend / dT + 0.5)
```

```
Amp = Val(txtAmp.Text)
```

```
A0max = A0max * Amp
```

```
txtUnit.Text = Unit
```

```
txtA0max.Text = A0max
```

```
txtdT.Text = dT
```

```
txtTstep.Text = Tstep
```

```
Acc(0) = 0
```

```
For i = 1 To Tstep
```

```
Input #1, Acc(i)
```

```
Acc(i) = Acc(i) * Amp
```

```
Next i
```

```
Close #1
```

```
Exit Sub
```

```
OpenError:
```

```
MsgBox "地震記録データを読み込めませんでした", , "エラー"
```

```
End
```

```
End Sub
```

```
*****
```

```
Private Sub 数値積分()
```

```
*****
```

```
Dacc = Acc(istep) - Acc(istep - 1)
```

```
KB = K + C / (2 * bet * dT) + M / (bet * dT * dT)
```

```
PB = M * (-Dacc + V0 / (bet * dT) + A0 / (2 * bet)) + C * (V0 / (2 * bet) + A0 * dT * (1 / (4 * bet) - 1))
```

```
DX = PB / KB
```

```
DV = DX / (2 * bet * dT) - V0 / (2 * bet) - A0 * dT * (1 / (4 * bet) - 1)
```

```
DA = DX / (bet * dT * dT) - V0 / (bet * dT) - A0 / (2 * bet)
```

```
X = X0 + DX
```

```
V = V0 + DV
```

```
A = A0 + DA
```

```
End Sub
```

```
*****
```

```
Private Sub グラフ初期設定()
```

```
*****
```

```
Dim i&
```

```
Dim clBlue As Long
```

```
Dim Xaxis#, Wi#
```

```
clBlue = QBColor(1)
```

```
グラフ1.Cls
```

```
Const 幅 As Single = 1000: Const 高さ As Single = 1000
```

```
グラフ1.Scale (-100, 高さ + 50)-(幅 + 100, -(高さ + 50)) 'スケールの設定
```

```
*** グラフの枠
```

```
グラフ1.DrawStyle = 0 'Solid Line
```

```
グラフ1.Line (0, 0)-(幅, 0), clBlue 'X-軸
```

```
グラフ1.Line (0, 高さ)-(幅, 高さ), clBlue 'X-軸
```

```
グラフ1.Line (0, -高さ)-(幅, -高さ), clBlue 'X-軸
```

```
グラフ1.Line (0, -高さ)-(0, 高さ), clBlue 'Y-軸
```

```
グラフ1.Line (幅, -高さ)-(幅, 高さ), clBlue 'Y-軸
```

```
*** y軸の目盛り
```

```
グラフ1.DrawStyle = 2 'Dot Line
```

```
Xaxis = Val(txtXaxis.Text)
```

```
Hscl = 高さ / Xaxis
```

```

For i = 1 To 4
    グラフ1.Line (0, i * (高さ / 5))-(幅, i * (高さ / 5)), clBlue
    グラフ1.Line (0, -i * (高さ / 5))-(幅, -i * (高さ / 5)), clBlue
Next i

'*** X軸目盛り(5秒間隔)
Wsc1 = 幅 / Tstep
W1 = 幅 * (1 / Tend)
i = 0
10 i = i + 1
Wi = Int(i * W1)
If Wi < 幅 Then
    グラフ1.Line (Wi, -高さ)-(Wi, 高さ), clBlue
    GoTo 10
End If

' 文字表示
グラフ1.CurrentX = -20: グラフ1.CurrentY = -20: グラフ1.Print "0"
グラフ1.CurrentX = 1000: グラフ1.CurrentY = -20: グラフ1.Print Tend
グラフ1.CurrentX = 1000: グラフ1.CurrentY = -200: グラフ1.Print "時間(sec)"
グラフ1.CurrentX = -100: グラフ1.CurrentY = 800: グラフ1.Print "変位(cm)"
グラフ1.PSet (0, X0 * 100 * Hsc1)

End Sub

'*****
Private Sub グラフ()
'*****

    グラフ1.DrawStyle = 0      'Solid Line
    グラフ1.Line -(istep * Wsc1, X * 100 * Hsc1), QBColor(12)

End Sub

'*****
Private Sub 質点初期設定()
'*****
    Dim clBlue As Long

    Pic質点.Cls

    Const 幅 As Single = 100: Const 高さ As Single = 100

    clBlue = QBColor(1)

    Pic質点.Scale (-(10 + 幅), 高さ + 30)-(幅 + 10, -10)      'スケールの設定

    Pic質点.DrawStyle = 0      'Solid Line
    Pic質点.Line (-幅, 0)-(幅, 0), clBlue                      'X-軸
    Pic質点.Line (0, 0)-(0, 高さ), clBlue                      '構造物

End Sub

'*****
Private Sub 質点アニメ()
'*****

    Const 幅 As Single = 100: Const 高さ As Single = 100
    Pic質点.Cls

    Pic質点.DrawStyle = 0      'Solid Line
    Pic質点.DrawWidth = 2
    Pic質点.Line (-幅, 0)-(幅, 0), QBColor(1)                  'X-軸
    Pic質点.Line (0, 0)-(X * 200, 高さ), QBColor(1)
    Pic質点.DrawWidth = 5
    Pic質点.Circle (X * 200, 高さ), 20, QBColor(12)

End Sub

'*****
Private Sub 出力ファイル初期設定()
'*****

    On Error GoTo OpenError

    OutFile = txtOutFile.Text
    OutFile = OutPath & "¥" & OutFile      '出力ファイル名
    Open OutFile For Output As #2

    Print #2, "固有周期T= "; T; "sec      減衰定数h= "; H; "%"
    Print #2, "地震記録ファイル: "; File1.FileName; "      入力倍率= "; Amp
    Print #2, " step 時間(s) 地動加速度 相対加速度 絶対加速度 速度(cm/sec) 変位(cm)"

```

```
Close #2

Exit Sub

OpenError:
MsgBox "出力ファイルの設定が間違っています", , "エラー"
End

End Sub

'*****
Private Sub データ出力()
'*****
Dim ステップ As String * 5, 時間 As String * 9
Dim 地動加速度 As String * 12, 相対加速度 As String * 12, 絶対加速度 As String * 12
Dim 速度 As String * 12, 変位 As String * 12

Open OutFile For Append As #2

RSet ステップ = CStr(Format(istep, "0"))
RSet 時間 = CStr(Format(istep * dT, "0.00"))
RSet 地動加速度 = CStr(Format(Acc(istep) * 100, "0.0"))
RSet 相対加速度 = CStr(Format(A * 100, "0.0"))
RSet 絶対加速度 = CStr(Format((A + Acc(istep)) * 100, "0.0"))
RSet 速度 = CStr(Format(V * 100, "0.00"))
RSet 変位 = CStr(Format(X * 100, "0.000"))

Print #2, ステップ; 時間; 地動加速度; 相対加速度; 絶対加速度; 速度; 変位

Close #2

End Sub

Private Sub Drive1_Change()
' カレントドライブの選択
Dir1.Path = Drive1.Drive
End Sub

Private Sub Dir1_Change()
' カレントパスの選択
File1.Path = Dir1.Path
End Sub

Private Sub Drive2_Change()
' カレントドライブの選択
Dir2.Path = Drive2.Drive
End Sub

Private Sub Dir2_Change()
' カレントパスの選択
OutPath = Dir2.Path
End Sub

Private Sub Timer1_Timer()
経過時間 = 経過時間 + 0.01
End Sub
```

```
*****
```

```
** 変数の説明
```

```
*****
```

```
*** 解析モデル
```

```
Public T As Double ' * 固有周期T
Public M As Double ' * 質量M
Public K As Double ' * 剛性K
Public H As Double ' * 減衰定数h
Public C As Double ' * 減衰係数C
Public j As Double
```

```
*** 地震記録
```

```
Public EqFile As String ' * 地震記録のファイル名
Public EqName As String ' * 地震記録の説明 (ファイルの1行目データ)
Public dT As Double ' * 地震記録の時間刻み (ファイルの2行目のデータ)
Public Unit As String ' * 地震記録の単位 (ファイルの3行目のデータ)
Public A0max# ' * 加速度の最大値 (ファイルの4行目のデータ)
Public Acc(100000) As Double ' * 加速度データ (ファイルの5行目以降)

Public Tend As Double ' * 解析の継続時間
Public Tstep% ' * 解析ステップ数
Public Amp As Double ' * 地震波の入力倍率
```

```
*** 数値積分
```

```
Public bet As Double ' * Newmark  $\beta$  法の  $\beta$ 
Public X, X0, DX As Double ' * 変位 (現ステップ、前ステップ、増分)
Public V, V0, DV As Double ' * 速度 (現ステップ、前ステップ、増分)
Public A, A0, DA As Double ' * 加速度 (現ステップ、前ステップ、増分)
Public Xmax#, Vmax#, Amax# ' * 変位、速度、絶対加速度の最大応答
Public Xmin#, Vmin#, Amin# ' * 変位、速度、絶対加速度の最小応答
Public DAcc As Double ' * 地動加速度増分
Public KB, PB As Double
```

```
*** 応答スペクトル用
```

```
Public Xmaxmax#, Vmaxmax#, Amaxmax# ' 最大応答値
Public Xmax2(2000), Vmax2(2000), Amax2(2000) As Double ' 応答スペクトル
Public j1, j2, j3 As Double ' 最大応答値の時のステップ
```

```
*** グラフ用
```

```
Public Wscl# ' * 1ステップの幅
Public W1# ' * 1秒分の幅
Public Hscl# ' * 1cmの高さ
Public l As Integer
```

```
*** 出力ファイル
```

```
Public OutFile As String ' * 出力ファイル名

Public i%, istep%
Public 経過時間#
Public dmy$
```

EL CENTRO 1940.5.18 IMPERIAL VALLEY EQ. (NS), GAL, 0.02SEC. JAPAN ARCHT CENTER

0.02

gal

341.7

-1.4	-10.8	-10.1	-8.8	-9.5	-12.0	-14.2
-12.8	-11.0	-8.5	-8.5	-13.1	-17.6	-19.4
-16.2	-14.4	-10.8	-8.2	-4.2	-6.6	-13.1
-19.0	-19.6	-6.6	3.0	14.1	-4.9	-12.8
-14.4	-20.3	-26.0	-32.5	-30.6	-17.2	-19.7
-16.3	-16.4	-6.7	2.5	15.0	23.6	25.2
33.6	46.3	49.2	41.9	35.9	27.1	23.5
33.9	41.2	53.0	63.9	73.2	65.2	59.9
40.0	40.0	6.3	-51.5	-78.7	-60.3	-48.4
-25.0	-5.9	13.4	30.8	49.9	71.0	99.5
121.9	152.9	144.9	115.5	93.5	89.2	92.6
83.9	90.1	99.3	120.9	32.8	-147.5	-206.6
-198.9	-203.4	-181.6	-172.5	-175.2	-175.3	-180.5
-163.0	-134.7	-108.7	-78.2	-42.9	-1.7	36.0
78.5	116.4	159.8	196.0	241.2	272.9	303.6
320.0	341.7	282.1	232.4	-119.8	-237.3	-164.0
-186.5	-109.5	-75.3	-17.3	11.3	53.3	89.5
118.6	175.7	57.6	-263.1	-154.7	-172.9	-101.2
-57.9	23.7	-67.0	-198.0	-164.1	-168.5	-148.1
-123.1	-100.1	-75.1	-52.3	-27.1	-4.4	18.8
-9.5	-43.3	-83.8	-95.1	-71.6	-59.9	-33.4
-10.8	18.5	42.0	67.3	-9.7	-37.2	-4.0
1.1	34.4	56.5	88.3	113.0	136.3	21.9
24.1	68.3	68.9	131.8	135.3	204.0	-93.1
-130.8	-69.2	-54.6	7.2	67.5	-106.7	-148.8
-107.1	-116.2	-76.2	-55.9	-21.5	-12.6	-67.4
-32.4	-33.7	-10.9	1.7	29.9	48.8	60.8
22.2	-3.2	-24.5	7.7	21.1	56.8	82.6
120.6	147.8	173.7	42.1	2.9	25.9	29.3
-5.5	-14.7	14.3	20.6	49.9	64.5	95.7
112.8	144.7	162.9	194.5	185.6	198.4	176.9
125.0	-120.7	-54.2	-38.4	-31.1	-111.8	-166.1
-246.4	-202.5	-183.5	-131.7	-96.0	-32.5	15.4
81.6	131.9	181.8	-5.8	-16.9	28.5	44.7
98.3	142.4	185.3	245.6	168.5	-138.0	-99.9
-108.9	-90.7	-46.9	-125.0	-211.1	-161.7	-169.2
-130.6	-111.1	-77.3	-51.0	-54.4	-120.0	-120.9
-115.8	-114.5	-71.7	-54.6	6.4	-80.4	-163.4
-85.9	-96.1	-39.6	-14.7	31.9	64.8	87.6
47.2	19.8	-2.7	29.2	44.5	78.5	103.3
135.2	160.6	186.1	128.1	64.0	20.4	31.4
37.3	49.6	23.5	-8.4	-16.8	-11.3	-22.9
-24.8	-15.7	-6.9	14.7	37.9	57.9	25.5
-4.1	-42.8	-13.3	9.5	23.0	-12.9	-5.0
8.0	21.0	38.0	51.0	15.7	-3.2	-11.1
0.5	7.6	3.5	-9.5	-3.6	-1.6	3.8
8.5	-5.6	-30.4	-42.1	-24.4	-23.6	-17.7
-12.9	-1.8	20.3	-10.8	-9.1	-3.4	-10.6
-11.1	-9.9	-0.2	7.3	23.5	35.5	70.5